Generators of $GL(2, \mathbb{F}_3)$ I came across the statement that for $GL(2, \mathbb{F}_3)$ we can choose generators $x$, $y$, $z$, that satisfy relations $x^4=y^3=z^2=xyz$, $(xyz)^2=1$. What matrices can I take for $x$, $y$, $z$?
 A: The statement is not true, though it is close.  You can check this using the small groups library from within the computer algebra system GAP:
gap> f := FreeGroup( "x", "y", "z" );;
gap> g := f / ParseRelators( GeneratorsOfGroup(f),
> "x^4 = y^3 = z^2 = xyz, (xyz)^2 = 1" );;
gap> IdGroup( g );
[ 48, 28 ]
gap> IdGroup( GL(2,3) );
[ 48, 29 ]
gap> StructureDescription(g/Intersection(DerivedSubgroup(g),Center(g)));
"S4"

Thus the group defined by that presentation is a group of order 48 that is a double cover of S4 ≅ PGL(2,3), but it is the "other" Schur cover, not GL(2,3).  One important difference between these groups is their 2-local structure.  The Sylow 2-subgroup of GL(2,3) is a quasi-dihedral group of order 16, while the Sylow 2-subgroup of the "other" group is quaternion of order 16.  This might be easier to see at the element level: GL(2,3) has two different conjugacy classes of involutions (elements of order 2), while the "other" group has a unique class of involutions (both of these statements are also true for their Sylow 2-subgroups).
If you want matrices for this "other" group, you can again use GAP:
# In GAP 4.5:
gap> g := SchurCoverOfSymmetricGroup(4,3,1);;
# In GAP 4.4:
gap> g := Group([ [ [ 0*Z(3), Z(3) ], [ Z(3)^0, Z(3^2)^6 ] ],
> [ [ Z(3^2)^2, 0*Z(3) ], [ 0*Z(3), Z(3^2)^6 ] ] ]);;
# In either:
gap> xyz := Filtered( Tuples(g,3),
>  function(xyz)
>    local x,y,z;
>    x:=xyz[1];
>    y:=xyz[2];
>    z:=xyz[3];
>    return x^4 = y^3 and y^3 = z^2 and z^2 = x*y*z and (x*y*z)^2 = x^0
>    and Subgroup( g, xyz ) = g;
>  end );;
gap> orb := Orbits( AutomorphismGroup(g), xyz, OnTuples );;
gap> Size(orb);
1

so there is only one such generating set up to isomorphism.  We can display it in GAP:
gap> Display(orb[1][1][1]);
gap> Display(orb[1][1][2]);
gap> Display(orb[1][1][3]);

But TeX is a bit nicer:
$$
x = \left(\begin{array}{rr}0&1\\-1&\sqrt{2}\end{array}\right), \quad
y = \left(\begin{array}{rr}1&\sqrt{-1}\\\sqrt{-1}&0\end{array}\right), \quad
z = \left(\begin{array}{rr}\sqrt{-1}&0\\-1+\sqrt{-2}&-\sqrt{-1}\end{array}\right), \quad
$$
where √2 ≡ −√−1 mod 3, and √−2 ≡ 1 mod 3.
This defines a subgroup of GL(2,9) that is isoclinic but not isomorphic to GL(2,3).  Multiplying x and z by √−1 gives an isomorphic copy of GL(2,3).
A: I just remark that ${\rm GL}(2,3)$ has a presentation $\langle x,y,z : x^8 = y^3 = z^2 = (xyz) = [x^4,y ] = 1 \rangle.$ This presentation gives a double cover of $S_4,$ but one which contains more than one involution, since $z \neq x^4.$
You can take $z = \left( \begin{array}{crclcr} 1 & 0\\ 0 & 2 \end{array} \right)$ and $y = \left( \begin{array}{clcr} 2 & 2\\1 & 0 \end{array} \right) .$
